
ABSTRACT: This note introduces Rees algebras and some of their uses, with
illustrations from version 2.2 of the Macaulay2 package ReesAlgebra.m2.

INTRODUCTION. A central construction in modern commutative algebra starts
from an ideal I in a commutative ring R, and produces the Rees algebra

R(I ) := R⊕ I ⊕ I 2
⊕ I 3
⊕ · · · ∼= R[I t] ⊂ R[t],

where R[t] denotes the polynomial algebra in one variable t over R. For basics
on Rees algebras, see [Vasconcelos 1994] and [Swanson and Huneke 2006], and
for some other research, see [Eisenbud and Ulrich 2018; Kustin and Ulrich 1992;
Ulrich 1994], and [Valabrega and Valla 1978].

From the point of view of algebraic geometry, the Rees algebra R(I ) is a homo-
geneous coordinate ring for the graph of a rational map whose total space is the
blowup of Spec R along the scheme defined by I. (In fact, the “Rees algebra” is
sometimes called the “blowup algebra”.)

Rees algebras were first studied in the algebraic context by David Rees, in the
now-famous paper [Rees 1958]. Actually, Rees mainly studied the ring R[I t, t−1

],
now also called the extended Rees algebra of I.

Mike Stillman and I wrote a Rees algebra script for Macaulay classic. It was aug-
mented, and made into the [Macaulay2] package ReesAlgebra.m2 around 2002,
to study a generalization of Rees algebras to modules described in [Eisenbud et al.
2003]. Subsequently Amelia Taylor, Sorin Popescu, the present author, and, at the
Macaulay2 Workgroup in July 2017, Ilir Dema, Whitney Liske, and Zhangchi Chen
contributed routines for computing many of the invariants of an ideal or module
defined in terms of Rees algebras. These routines comprise the package’s primary
utility, since Rees algebras of modules other than ideals are comparatively little
studied.

The author is grateful to the National Science Foundation for partial support.
MSC2010: primary 13A30, 13B22, 13D02; secondary 14C17, 14E15.

We first describe the construction and an example from [Eisenbud et al. 2003].
Then we list some of the functionality the package now has and illustrate it with
a theorem of Morey and Ulrich. Finally we give examples of how Rees algebras
appear in the Fulton–MacPherson intersection theory and in the resolution of sin-
gularities.

1. THE REES ALGEBRA OF A MODULE. There are several possible ways of ex-
tending the Rees algebra construction from ideals to modules. For simplicity we
will henceforward only consider finitely generated modules over Noetherian rings.
Huneke and Ulrich and I argued in [Eisenbud et al. 2003] that the most natural way
to extend the definition is to think of R[I t] as the image of the map of symmetric
algebras Sym(φ) : SymR(I )→ SymR(R)= R[t], and to generalize it to the case
of an arbitrary finitely generated module M by setting

R(M)= image Sym(φ),

where φ is a versal map from M to a free module. Such a versal map may be
computed as the composition of the diagonal embedding

M→⊕m
i=1 M,

with the map
⊕

m
i=1φi : ⊕

m
i=1 M→ Rm,

where φ1, . . . , φm generate HomR(M, R).
Though this is not immediate, the Rees algebra of an ideal in a Noetherian ring,

in this sense, is the same as the Rees algebra in the classical sense, and in most cases
one can take any embedding of the module into a free module in the definition:

Theorem 1.1 [Eisenbud et al. 2003, Theorems 0.2 and 1.4]. Let R be a Noetherian
ring and let M be a finitely generated R-module. Let φ : M→ G be a versal map
of M to a free module. Suppose that φ is an inclusion, and let ψ : M→ G ′ be any
inclusion of M into a free module G ′. If R is torsion-free over Z or R is unmixed
and generically Gorenstein or M is free locally at each associated prime of R,
or G ′ = R, then the image of Sym(φ) and the image of Sym(ψ) are naturally
isomorphic.

Nevertheless some examples do violate the conclusion of Theorem 1.1. Here is
one from [Eisenbud et al. 2003] in characteristic 5 (any finite characteristic would
work similarly).
i1 : p = 5;
i2 : R = ZZ/p[x,y,z]/(ideal(x^p,y^p)+(ideal(x,y,z))^(p+1));
i3 : M = module ideal(z);

It is easy to check that M ∼= R1/(x, y, z)p. We write ι : M→ R1 for the embedding
as an ideal and ψ for the embedding M→ R2 sending z to the vector (x, y).

i4 : iota = map(R^1,M,matrix{{z}});
i5 : psi = map(R^2,M,matrix{{x},{y}});

Finally, we choose a versal embedding M→ R3. It sends z to the vector (x, y, z):

i6 : phi = versalEmbedding(M);

We now compute the kernels of the three maps on symmetric algebras:

i7 : Iiota = symmetricKernel iota;
i8 : Ipsi = symmetricKernel psi;
i9 : Iphi = symmetricKernel phi;

and check that the ones corresponding to φ and ι are equal, whereas the ones
corresponding to ψ and φ are not — they differ in degree p.

i10 : Iiota == Iphi
o10 = true
i11 : Ipsi == Iphi
o11 = false
i12 : numcols basis(p,Iphi)
o12 = 3
i13 : numcols basis(p,Ipsi)
o13 = 1

2. THE REES ALGEBRA AND ITS RELATIONS. The central routine, reesIdeal
(with synonym: reesAlgebraIdeal), computes an ideal defining the Rees algebra
R(M) as a quotient of a polynomial ring over R from a free presentation of M.
From the Rees ideal we immediately get reesAlgebra M. In the case when M
is an ideal in R we also compute the important associatedGradedRing M =
R(M)/M (and the more geometric sounding but identical normalCone M ). If I
is a (homogeneous) ideal primary to the maximal ideal of a standard graded ring R
we compute the Hilbert–Samuel multiplicity of I with the routine multiplicity.

We now describe the basic computation. Suppose that M has a set of generators
represented by a map from a free module,

F α
−→M→ 0,

and suppose F = Rn. The symmetric algebra of F over R is then a polynomial
ring SymR(F)= R[t1, . . . , tn] on n new indeterminates t1, . . . , tn . By the universal
property of the symmetric algebra there is a canonical surjection SymR(F)→
SymR(M), so we may compute the Rees algebra of M as a quotient of SymR(F).
The expression

I = reesIdeal M

first uses versalEmbedding M to compute a versal map from M to a free module
β : M→ G. The expression symmetricKernel α ◦β then constructs the map of

symmetric algebras β ◦α : SymR(F)→ SymR(G) and uses the built-in Macaulay2
routine to compute the kernel

I = reesIdeal M = ker Sym(β ◦α) : SymR(F)→ SymR(G).

There is a different way of computing the Rees algebra that is often much more
efficient. It begins by constructing the symmetric algebra of M, and uses the obser-
vation that the construction of the Rees algebra commutes with localization. See
[Eisenbud 1995, Appendix 2] for the necessary facts about symmetric algebras.

Suppose that M has a free presentation,

G φ
−→ F α

−→M→ 0.

The right exactness of the symmetric algebra functor implies that the symmetric
algebra of M is the quotient of SymR(F) by an ideal I0 that is generated by the
entries of the matrix

(t1 · · · tn) ◦φ,

(where we have identified φ with SymR(F)⊗R φ). Thus I0 is generated by poly-
nomials that are linear in the variables ti (and because M is the degree 1 part of
R(M), these are the only linear forms in the ti in the Rees ideal).

If f ∈ R is an element such that M[ f −1
] is free on generators g1, . . . , gn , it

follows that after inverting f , the Rees algebra of M becomes a polynomial ring
over R[ f −1

] on indeterminates corresponding to the gi :

R(M)[ f −1
] = SymR(M[ f

−1
])= R[G1, . . . ,Gn].

Now suppose in addition that f is a non-zerodivisor in R. In the diagram

SymR(F)
α

//

��

SymR(M)
β

//

��

SymR(G)

��

SymR(F)[ f
−1
]

α
// SymR(M)[ f

−1
]

β
// SymR(G)[ f

−1
]

the two outer vertical maps are inclusions, and it follows that the Rees ideal, which
is the kernel of the map R(F)= SymR(F)→R(M), is equal to the intersection
of R(F) with the kernel of

SymR(F)[ f
−1
]

β
−→ SymR(G)[ f

−1
].

This intersection may be computed as I0 : f∞. The command

reesIdeal(I, f )

computes the Rees ideal in this way.
More generally, we say that a module N is of linear type if the Rees ideal of M

is equal to the ideal of the symmetric algebra of M ; for example, any complete

intersection ideal is of linear type, and the condition can be tested by the command

isLinearType M.

The procedure above really requires only that f be a non-zerodivisor in R and that
M[ f −1

] be of linear type over R[ f −1
].

3. REDUCTIONS AND THE SPECIAL FIBER. A reduction J of an ideal I is a
subideal J ⊂ I over which I is integrally dependent. In concrete terms this means
that there is some integer r such that J I r

= I r+1, and the minimal r with this
property is called the reduction number. The property of being a reduction is tested
by isReduction I, and reductionNumber I computes the reduction number.

Now suppose that m is a maximal ideal containing I. The special fiber ring is
by definition R(I )/mR(I ). It is a standard graded algebra over the field k := R/m,
a quotient of SymR(F)/m= k[t1, . . . , tn] where, as before, F is a free module of
rank n with a surjection to M. The defining ideal of the special fiber ring, and the
ring itself, are computed using specialFiberIdeal I and specialFiberRing I.

The dimension of the special fiber ring is called the analytic spread of I, usually
denoted

`(I )= analyticSpread I.

Northcott and Rees [1954] proved that if k is infinite then there always exist re-
ductions generated by `(I ) elements, and this is the minimum possible number;
these are called minimal reductions. The smallest possible reduction number for I
with respect to a minimal reduction is by definition reductionNumber I. (This is
always achieved by any ideal generated by `(I ) sufficiently general scalar linear
combinations of the generators of I ; but note that when I is homogeneous but has
generators of different degrees such linear combinations are sometimes necessarily
inhomogeneous.)

An interesting special case occurs when R is a graded ring over k = R0 and the
generators g1, . . . , gn of I are all homogeneous of the same degree. In this case
the special fiber ring is easily seen to be equal to the subring k[g1, . . . , gn] (usually
not a polynomial ring) generated by the elements gi .

For example, if I is the ideal of p× p minors of a p× (p+ q) matrix, then the
special fiber ring is equal to the homogeneous coordinate ring G of the Grassman-
nian of p-planes in p+ q space. It follows that `(I ) = dim G = pq + 1, and the
reduction number of I is (p− 1)(q − 1).

4. FINDING ELEMENTS OF THE REES IDEAL. Let M be an R-module and let φ :
Rs
→ Rm be its presentation matrix. We identify SymR(R

m) with the polynomial
ring R[t1, . . . , tm]. By the universality of the symmetric algebra construction, the

symmetric algebra of I has the form

SymR(I )= R[t1, . . . , tm]/(Tφ),

where we have written T for the vector (t1 . . . tm) ∈ R[t1, . . . , tm]m, whose entries
correspond to the generators of I, and written (Tφ) for the ideal generated by the
entries of the product

(t1 · · · tm)φ.

If J := (x1, . . . , xn)⊂ R is an ideal containing I, and we write

X = (x1 · · · xn) ∈ R[t1, . . . , tm]n,

then there is a matrix ψ defined over R[t1, . . . , tm], called the Jacobian dual of φ
with respect to X, such that Tφ = Xψ . (The matrix ψ is generally not unique;
Macaulay2 computes it using Gröbner division with remainder.)

If I, J each contain a non-zerodivisor then J will have grade ≥ 1 on the Rees
algebra R(I ). Since (Tφ) is contained in the defining ideal of the Rees algebra,
the vector X is annihilated by the matrix ψ when regarded over the Rees algebra,
and the relation Xψ ≡ 0 in R(I ) implies that the m ×m minors of ψ are in the
Rees ideal of I.

In very favorable circumstances, one may even have the equality

reesIdeal I == ideal(Tφ)+ minors(m, ψ).

We illustrate with a theorem of Morey and Ulrich. Recall that an ideal I is said
to satisfy the condition G` if the number of generators of the localized ideal IP

is ≤ codim P for every prime ideal P of codimension < `; equivalently, if I has
presentation matrix φ as above,

codim Im−p(φ) > p
for 1≤ p < `.

Theorem 4.1 [Morey and Ulrich 1996]. Let R be a local Gorenstein ring with
infinite residue field, let I be a perfect ideal of grade 2 with m generators, let φ be
the presentation matrix of I, and let ψ be the Jacobian dual matrix. Let `= `(I )
be the analytic spread. Suppose that I satisfies the condition G`. The following
conditions are equivalent:

(1) R(I ) is Cohen–Macaulay and I(m−`)(φ)= I1(φ)
m−`.

(2) r(I ) < ` and Im+1−`φ = (I1φ)
m+1−`.

(3) The ideal of R(I ) is equal to the sum of the ideal of Sym(I ) with the Jacobian
dual minors, Imψ .

We can check all these conditions with functions in the package. We start with
the presentation matrix φ of an m=n+1-generator perfect ideal such that the first

row consists of the n variables of the ring, and the rest of the rows are reasonably
general (in this case random quadrics):

i2 : setRandomSeed 0
i3 : n=3;
i4 : kk = ZZ/101;
i5 : S = kk[a_0..a_(n-2)];
i6 : phi = transpose map(S^(n-1),S^{-1,(n-1):-2},

(i,j) -> if j == 0 then a_i else random(2,S));
3 2

o6 : Matrix S <--- S
i7 : I = minors(n-1,phi);

This is a perfect codimension 2 ideal, as we see from the Betti table:

i8 : betti (F = res I)
0 1 2

o8 = total: 1 3 2
0: 1 . .
1: . . .
2: . 2 .
3: . 1 2

We compute the analytic spread ` and the reduction number r :

i12 : ell = analyticSpread I
o12 = 2
i13 : r = reductionNumber(I, minimalReduction I)
o13 = 1

Now we can check the condition G`, first probabilistically:

i15 : whichGm I >= ell
o15 = true

and now deterministically:

i17 : apply(toList(1..ell-1),
p-> {p+1, codim minors(n-p, phi)})

o17 = {{2, 2}}

We now check the three equivalent conditions of the Morey–Ulrich theorem. Since
` = n − 1 in this case, the second parts of conditions (1) and (2) are vacuously
satisfied, and since r < ` the conditions must all be satisfied. We first check that
R(I ) is Cohen–Macaulay:

i19 : reesI = reesIdeal I;
o19 : Ideal of S[w , w , w ]

0 1 2
i20 : codim reesI
o20 = 2

i21 : betti res reesI
0 1 2

o21 = total: 1 3 2
0: 1 . .
1: . . .
2: . 2 .
3: . 1 2

Finally, we wish to see that reesIdeal I is generated by the ideal of the symmetric
algebra together with the Jacobian dual:

i23 : psi = jacobianDual phi;
2 2

o23 : Matrix (S[w , w , w ]) <--- (S[w , w , w ])
0 1 2 0 1 2

We now compute the ideal J of the symmetric algebra; we do this by hand, since
the command symmetricAlgebra I would return the ideal over a different ring.

i25 : ST = ring psi
i26 : T = vars ST
o26 = | w_0 w_1 w_2 |
i27 : J = ideal(T*promote(phi, ST))
i28 : betti res J

0 1 2
o28 = total: 1 2 1

0: 1 . .
1: . . .
2: . 2 .
3: . . .
4: . . 1

i29 : J1 = minors(ell, psi)

We compute the resolution of G := J + J1, to see that the resulting ideal is perfect,
which also shows that it is the full ideal of the Rees algebra. We also check directly
that it has the same resolution as the computed Rees ideal of I :

i30 : betti (G = res trim (J+J1))
0 1 2

o30 = total: 1 3 2
0: 1 . .
1: . . .
2: . 2 .
3: . 1 2

i31 : betti res reesIdeal I
0 1 2

o31 = total: 1 3 2
0: 1 . .
1: . . .
2: . 2 .
3: . 1 2



Eisenbud :::: The ReesAlgebra package in Macaulay2 57

5. DISTINGUISHED SUBVARIETIES. The key construction in the Fulton–MacPherson
definition of the refined intersection product [Fulton 1998, Section 6.1] involves
normal cones, and is easy to implement using the tools in this package. The sim-
plest case is the intersection of two subvarieties X, V ⊂ Y. If X and V meet in the
expected dimension, defined to be dim V − codimY X, and the ambient variety Y
is smooth, then one can assign multiplicities mi to the components Wi of X ∩ V,
and the intersection product has the form [X ][V ] =

∑
mi [Wi ]. The astonishing

result of the Fulton–MacPherson theory is that if X ⊂ Y is locally a complete
intersection, then, no matter how singular Y and no matter how strange the actual
intersection X ∩ V, the intersection product X · V can be given a meaning as a
rational equivalence class of cycles of the expected dimension on X, or even on
certain distinguished subvarieties Zi of X ∩ V. This class comes with a canonical
decomposition

∑
i miαi , where the mi are positive integers, and αi is a cycle of the

expected dimension (possibly 0) on Zi ⊂ X ∩ V (the same Zi can appear several
times, with different multiplicities and cycles).

In the general case, the subvariety V is replaced by a morphism f : V → Y from
a variety V, and this is the key to the functoriality of the intersection product. The
routines in this package work in the general setting, but for simplicity we will stick
with the basic case in this description.

We now describe the distinguished subvarieties and their multiplicities. This
part of the construction sheafifies, so (as in the package) we work in the affine
case. We do not require any hypothesis on X, Y or V.

Let S be a ring (for example, the coordinate ring of Y ) and let I ⊂ S be an ideal
(for example, the ideal of X ). Write

T := grI S = S/I ⊕ I/I 2
⊕ · · ·

for the associated graded ring of I, and let π be the inclusion of S/I into T as the
degree 0 part.

Let f : S→ R be a ring homomorphism (for example, representing the projection
S→ S/(I (V ))). Let K ⊂ T be the kernel of the induced map grI S→ gr f (I )R R.

Let P1, . . . , Pm be the minimal primes over K in grI R. We define pi to be the
degree 0 part of Pi ; that is, pi := Pi ∩ S/I . These are the distinguished prime ideals
of S/I , and they clearly contain the kernel of f : S/I → R/ f (I )R, so in the case
where R = S/J they contain I + J. Thus, in this case, they represent subvarieties
of X ∩ V.

Let mi be the multiplicity with which Pi appears in the primary decomposition
of K — that is,

mi := lengthκ(Pi )
Pi Pi /K Pi ,

where κ(Pi )= TPi /PiPi
is the residue field at Pi . Returning to geometric language,

and the case where X ⊂ Y is locally a complete intersection in a quasiprojective

variety, the cycle class αi in the Chow group of the variety Zi corresponding to pi

is defined as the Gysin image of the class of the subvariety corresponding to Pi

in the projectivized normal bundle of X in Y — a construction not included in this
package.

Here are some simple examples in which distinguished is used to compute
the distinguished varieties of intersections in An, via the function intersectInP.
First, the familiar multiplicity 2 intersection of a conic with a tangent line.
i2 : kk = ZZ/101;
i3 : P = kk[x,y];
i4 : I = ideal"x2-y";J=ideal y;
i6 : intersectInP(I,J)
o6 = {{2, ideal (y, x)}}

Slightly more interesting, the following shows what happens when the intersections
aren’t rational:
i7 : I = ideal"x4+y3+1";
i8 : intersectInP(I,J)

2 2
o8 = {{1, ideal (y, x + 10)}, {1, ideal (y, x - 10)}}

The real interest in the construction is in the case of improper intersections. Here
are some typical results:
i9 : I = ideal"x2y";J=ideal"xy2";
i11 : intersectInP(I,J)
o11 = {{2, ideal x}, {5, ideal (y, x)}, {2, ideal y}}
i12 : intersectInP(I,I)
o12 = {{1, ideal y}, {4, ideal x}, {4, ideal (y, x)}}

6. REES ALGEBRAS AND DESINGULARIZATION. We conclude with an example
illustrating a general result about projective birational maps of varieties. Recall
that a map B → X of varieties is projective if it is the composition of a closed
embedding B ⊂ X ×Pn with the projection to X. It is birational if it is generically
an isomorphism. The inclusion of a ring into the Rees algebra of an ideal corre-
sponds to a map from Proj of the Rees algebra to Spec of the ring, called a blowup,
that is such a proper birational transformation, and in fact every proper birational
transformation to an affine variety (or more generally to any scheme, if one works
with sheaves of ideals) can be realized in this way.

The theorem of embedded resolution of singularities (proven by Hironaka in
characteristic 0 and conjectured in general) says that, given any subvariety X of a
smooth variety Y, there is a finite sequence of blowups

Bn→ · · · → B2→ B1→ Y

of smooth subvarieties that lie over the singular set of X, and a component of the
preimage of X in Bn that is smooth and maps birationally to X. In the case of

plane curves, this can be done with a sequence of blowups of closed points. But
in fact any sequence of blowups of a quasiprojective variety can be replaced with
a single blowup [Hartshorne 1977, Theorem II.7.17] of a more complicated ideal.
We illustrate this with the desingularization of a tacnode (the union of two smooth
curves that meet with a simple tangency).

Example 6.1. Blowing-up (x2, y) in k[x, y] desingularizes the tacnode x2
− y4 in

a single step.

i1 : R = ZZ/32003[x,y];
i2 : tacnode = ideal(x^2-y^4);
i3 : mm = ideal(x,y^2);
i4 : B = first flattenRing reesAlgebra mm;
i5 : irrelB = ideal(w_0,w_1);
i6 : proj = map(B,R,{x,y});
i7 : totalTransform = proj tacnode

4 2
o7 = ideal(- y + x )
i8 : netList (D = decompose totalTransform)

+-----------------------+
o8 = |ideal (y, x) |

+-----------------------+
| 2 |
|ideal (y + x, w + w )|
| 0 1 |
+-----------------------+
| 2 |
|ideal (y - x, w - w )|
| 0 1 |
+-----------------------+

i9 : exceptional = proj mm
2

o9 = ideal (x, y )
i10 : strictTransform = saturate(

totalTransform, exceptional);

i11 : netList decompose strictTransform
+-----------------------+
| 2 |

o11 = |ideal (y + x, w + w )|
| 0 1 |
+-----------------------+
| 2 |
|ideal (y - x, w - w )|
| 0 1 |
+-----------------------+

i12 : sing0 = sub(ideal singularLocus strictTransform, B);
i13 : sing = saturate(sing0,irrelB)
o13 = ideal 1

The last line asserts that the singular locus of the strict transform is empty; that is,
the scheme defined by strictTransform is smooth (in this case it is the union of
two disjoint smooth curves).

SUPPLEMENT. The online supplement contains version 2.2 of ReesAlgebra.m2.

